AD-A032 910 MISSOURI UNIV KANSAS CITY DEPT OF MATHEMATICS F/G 12/1
ALGORITHMS FOR RATIONAL APPROXIMATIONS FOR A CONFLUENT HYPERGEO--ETC(U)
SEP 76 Y L LUKE AF-AFOSR-2520-73
UNCLASSIFIED AFOSR-TR-76-1165 NL

1 OF 1
ADA032910



END
DATE
FILMED
1 -77

4

AFOSR - TR- 76 - 1165

Algorithms for Rational Approximations for A Confluent Hypergeometric Function II*

by

Yudell L. Luke
Department of Mathematics
University of Missouri
Kansas City, Missouri

*This research work was sponsored by the Air Force Office of Scientific Research under Grant 73-2520.

UNCLASSIFIED
SECURITY CLASSIFICATION OF THIS PAGE (When Data Entered)

REPORT DOCUMENTATION PAGE

READ INSTRUCTIONS BEFORE COMPLETING FORM

1. REPORT NUMBER: AFOSR - TR - 76 - 1165	2. GOVT ACCESSION NO.
3. RECIPIENT'S CATALOG NUMBER	
4. TITLE (and Subtitle): ALGORITHMS FOR RATIONAL APPROXIMATIONS FOR A CONFLUENT HYPERGEOMETRIC FUNCTION II	5. TYPE OF REPORT & PERIOD COVERED: Interim rept.
	6. PERFORMING ORG. REPORT NUMBER
7. AUTHOR(s): Yudell L. Luke	8. CONTRACT OR GRANT NUMBER(s): AF-AFOSR-2520-73
9. PERFORMING ORGANIZATION NAME AND ADDRESS: Department of Mathematics, University of Missouri, Kansas City, Missouri 64110	10. PROGRAM ELEMENT, PROJECT, TASK AREA & WORK UNIT NUMBERS: 61102F, 9749-03
11. CONTROLLING OFFICE NAME AND ADDRESS: Air Force Office of Scientific Research/NM, Bolling AFB, Washington, D.C. 20332	12. REPORT DATE: September 1, 1976
	13. NUMBER OF PAGES
14. MONITORING AGENCY NAME & ADDRESS (if different from Controlling Office): 1 Sep 76, 36p.	15. SECURITY CLASS. (of this report): UNCLASSIFIED
	15a. DECLASSIFICATION/DOWNGRADING SCHEDULE

16. DISTRIBUTION STATEMENT (of this Report)

Approved for public release; distribution unlimited.

17. DISTRIBUTION STATEMENT (of the abstract entered in Block 20, if different from Report)

18. SUPPLEMENTARY NOTES

19. KEY WORDS (Continue on reverse side if necessary and identify by block number)

Confluent Hypergeometric Function
Rational Approximation
Padé Approximation
Algorithm
FORTRAN Programs

20. ABSTRACT (Continue on reverse side if necessary and identify by block number)

This is a sequel to a previous paper where rational approximations for the confluent hypergeometric function $z^a U(a;c;z)$ were treated. Here we take up rational approximations for ${}_1F_1(a;c;-z)$. The confluent functions are very important in the

#20. (continued)

applications since they include as special cases the incomplete gamma function (special cases of which are exponential, sine and cosine integrals, Fresnel integrals and the error function), Bessel functions, parabolic cylinder functions and Coulomb wave functions. The subject of rational approximations for a wide class of functions including those named above were examined in some detail in my volumes on the special functions. In the special case where a is unity, the confluent function becomes an incomplete gamma function. In this event, complete a priori error analyses for the main diagonal Padé approximations and much more were presented. For general parameters, the rational approximations treated were not of the Padé class. It was shown that the rational approximations converge, but a complete a priori analysis was not available. One of the purposes of this report is to correct this deficiency. Further, FORTRAN programs are provided to evaluate the Padé and non-Padé rational approximations by using the appropriate recursion formulas to generate the numerator and denominator polynomials as a number, and to also evaluate the coefficients which define these polynomials. The programming of the routines was done for use by the IBM 370/168 operating under OS/VS Release 1.7 on the FORTRAN IV H-Extended Compiler, Release 2.1. All computer programs are written for quadruple precision and real arithmetic. By making a few simple changes, one can have double or single precision. Further, it is easy to get complex arithmetic along with any of the precisions noted above.

Summary

This is a sequel to a previous paper where rational approximations for the confluent hypergeometric function $z^a U(a;c;z)$ were treated. Here we take up rational approximations for ${}_1F_1(a;c;-z)$. The confluent functions are very important in the applications since they include as special cases the incomplete gamma function (special cases of which are exponential, sine and cosine integrals, Fresnel integrals and the error function), Bessel functions, parabolic cylinder functions and Coulomb wave functions. The subject of rational approximations for a wide class of functions including those named above were examined in some detail in my volumes on the special functions. In the special case where a is unity, the confluent function becomes an incomplete gamma function. In this event, complete a priori error analyses for the main diagonal Padé approximations and much more were presented. For general parameters, the rational approximations treated were not of the Padé class. It was shown that the rational approximations converge, but a complete a priori analysis was not available. One of the purposes of this report is to correct this deficiency. Further, FORTRAN programs are provided to evaluate the Padé and non-Padé rational approximations by using the appropriate recursion formulas to generate the numerator and denominator polynomials as a number, and to also evaluate the coefficients which define these polynomials. The programming of the routines was done for use by the IBM 370/168

operating under OS/VS Release 1.7 on the FORTRAN IV H-Extended Compiler, Release 2.1. All computer programs are written for quadruple precision and real arithmetic. By making a few simple changes, one can have double or single precision. Further, it is easy to get complex arithmetic along with any of the precisions noted above.

1. Introduction

In a previous paper, Luke (1), we considered rational approximations for the confluent hypergeometric function $z^a U(a;c;z)$. This is a sequel to the paper just noted in which we treat rational approximations for the confluent function

$$E(z) = {}_1F_1(a;c;-z), \tag{1}$$

$$E(z) = \sum_{k=0}^{\infty} \frac{(a)_k (-)^k z^k}{(c)_k k!} . \tag{2}$$

The connection between the confluent functions is given by

$$U(a;c;z) = \frac{\Gamma(1-c)}{\Gamma(b)}\, {}_1F_1(a;c;z) + \frac{\Gamma(c-1)}{\Gamma(a)}\, {}_1F_1(b;2-c;z), \quad b = 1+a-c. \tag{3}$$

The confluent functions are very important in the applications since they include as special cases the incomplete gamma functions (special cases of these are exponential, sine and cosine integrals, Fresnel integrals and error functions), Bessel functions, parabolic cylinder functions and Coulomb wave functions. For more details on all these transcendents, see Luke (2,3).

Consider a function $E(z)$,

$$E(z) = E_n(z) + R_n(z), \tag{4}$$

$$E_n(z) = A_n(z)/B_n(z) \tag{5}$$

where $A_n(z)$ and $B_n(z)$ are polynomials in z of degree n,

and $R_n(z)$ is the remainder. For each $E(z)$, with parameters specified, two FORTRAN programs along with illustrative examples are given. In the first, a positive integer N and z are also specified. The machine is given forms to evaluate sufficient initial values of $A_n(z)$ and $B_n(z)$, and computes subsequent values of $A_n(z)$ and $B_n(z)$ by means of a recursion formula and evaluates $E_n(z)$ for $n = 0,1,\ldots,N$. The machine prints $A_n(z)$, $B_n(z)$, $E_n(z)$, $E_{n+1}(z) - E_n(z)$ (that is, first differences) and $E_N(z) - E_n(z)$, all for $n = 0,1,\ldots,N$. The last two quantities can be viewed as a measure of the error with the latter preferred. This program is especially valuable both to get the desired approximants $E_n(z)$, and in the absence of applying the asymptotic forms of the remainder, to also get an appraisal of the value of n needed to achieve a given level of accuracy. Once n is known, it might be more convenient and economical to have the coefficients which define the polynomials in $A_n(z)$ and $B_n(z)$. This is furnished by the second program. Thus once z is specified, $A_n(z)$, $B_n(z)$ and $E_n(z)$ are readily found. The polynomials are especially valuable when one desires to use the approximation to simplify analytical formulas and to make further approximations such as the evaluation of integrals and transforms involving $E(z)$.

The technique for evaluating the coefficients in the numerator and denominator polynomials follows. It is convenient to treat more general forms than is required in the present analysis. Let

$$B_n(z) = L_n z^n {}_{q+f+3}F_{p+g+1}\left(\begin{matrix}-n, n+\lambda, \rho_q - a, c_f, 1\\ \beta+1,\ \alpha_p+1-a,\ d_g\end{matrix}\middle| -1/z\right) , \tag{6}$$

$$A_n(z) = L_n z^n \left[\frac{n(n+\lambda)(\rho_q - 1)c_f z}{(\beta+1)\alpha_p d_g} \right]^a$$

$$\times \sum_{k=0}^{n-a} \frac{(a-n)_k (n+\lambda+a)_k (\alpha_p)_k (c_f+a)_k}{(\beta+1+a)_k (\alpha_p+1)_k (d_g+a)_k k!}$$

$$\times {}_{q+f+3}F_{p+g+1}\left(\begin{matrix} -n+a+k, n+\lambda+a+k, \rho_q+k, c_f+a+k, 1 \\ \beta+1+a+k,\ \alpha_p+1+k,\ d_g+a+k \end{matrix} \middle| -1/z \right), \qquad (7)$$

or

$$A_n(z) = \left[\frac{n(n+\lambda)(\rho_q - 1)c_f}{(\beta+1)\alpha_p d_g} \right]^a L_n z^{n-a}$$

$$\times \sum_{k=0}^{n-a} \frac{(a-n)_k (n+\lambda+a)_k (\rho_q)_k (c_f+a)_k (-)^k z^{-k}}{(\beta+1+a)_k (\alpha_p+1)_k (d_g+a)_k}$$

$$\times {}_{p+q+f+2}F_{p+q+g+1}\left(\begin{matrix} -n+a+k, n+\lambda+a+k, \rho_q+k, c_f+a+k, \alpha_p \\ \beta+1+a+k,\ \alpha_p+1+k,\ d_g+a+k,\ \rho_q \end{matrix} \middle| 1 \right), \qquad (8)$$

where

$$L_n = \frac{(\beta+1)_n (\alpha_p+1-a)_n (d_g)_n}{(n+\lambda)_n n! (\rho_q - a)_n (c_f)_n}, \quad a = 0 \quad \text{or} \quad a = 1, \ \lambda = \alpha+\beta+1. \qquad (9)$$

It is easily proved that

$$B_n(z) = {}_{p+g+1}F^n_{q+f+1}\left(\begin{matrix} -n-\beta,\ -n-\alpha_p+a,\ 1-n-d_g \\ 1-\lambda-2n, 1-n-\rho_q+a, 1-n-c_f \end{matrix} \middle| (-)^r z \right),$$

$$= \sum_{k=0}^{n} u_k z^k, \ u_o = 1, \ r = p+q+f+g. \qquad (10)$$

Then the u_k's are readily evaluated by use of the recurrence formula

$$u_{k+1} = \frac{(-)^r(-n-\beta+k)(-n-\alpha_p+a+k)(1-n-d_g+k)u_k}{(1-\lambda-2n+k)(1-n-\rho_q+a+k)(1-n-c_f+k)(k+1)} . \tag{11}$$

Further, we can write

$$A_n(z) = \sum_{k=0}^{n-a} v_k z^k, \quad v_k = \sum_{m=0}^{k} t_m u_{k-m} ,$$

$$t_m = \frac{(-)^m(\alpha_p)_m}{(\rho_q)_m m!} , \tag{12}$$

where the u_k's are defined by (10) and are easily evaluated by (11). Since

$$t_{m+1} = -\frac{(\alpha_p+m)}{(\rho_q+m)(m+1)} t_m, \quad t_o = 1 , \tag{13}$$

we see that computation of v_k is direct.

In Luke (2,3), a is either 0 or 1, but in our present work $a = 0$. This a has nothing to do with the a in ${}_1F_1(a;c;-z)$. We now show how to get the forms in Sections 2 and 3. Put $\lambda = \beta+1$, $p = q = 1$. To get the Section 2 polynomials, let $\beta = f = g = 0$, $\alpha_1 = a$ and $\rho_1 = c$. To get the Section 3 polynomials, let $\beta = c-1$, take $f = g = 1$ and put $\alpha_1 = 1$, $\rho_1 = c$, $c_1 = 2$ and $d_1 = 1$.

As in our previous paper, the programs are written for certain IBM equipment in quadruple precision for real arithmetic. However, with a few minor changes, we can have single or double precision,

and for any precision we can also have complex arithmetic. All of this is well detailed in Luke (1, Section 4) and will not be repeated here.

2. Rational Approximations for ${}_1F_1(a;c;-z)$

Let

$$E(z) = {}_1F_1(a;c;-z) = e^{-z}\ {}_1F_1(c-a;c;z). \tag{1}$$

We suppose that neither a nor $c-a$ is a negative integer or zero, for otherwise from (1), $E(z)$ is a polynomial except for the possible presence of e^{-z}. We also suppose that $c \neq a$. This is for convenience only, because when $c = a$, $E(z) = {}_0F_0(-z) = e^{-z}$, and in this event one should use the rational approximations of Section 3 with $c = 1$ as they are simpler and more accurate. The Section 3 approximations with $c = 1$ can be found from those of this section by putting $c = a+1$ and letting $a \to \infty$. For general parameters, the rational approximations which follow are not of the Padé class. We write

$$E(z) = \{A_n(z)/B_n(z)\} + R_n(z) , \tag{2}$$

$$B_n(z) = L_n z^n\ {}_3F_1(-n,n+1,c;a+1;-1/z) \tag{3}$$

$$A_n(z) = L_n z^n \sum_{k=0}^{n} \frac{(-n)_k(n+1)_k(a)_k}{(a+1)_k(k!)^2}\ {}_4F_2\left(\begin{matrix}-n+k,n+1+k,c+k,1\\1+k,\ a+1+k\end{matrix}\middle|-1/z\right), \tag{4}$$

$$L_n = \frac{(a+1)_n}{(n+1)_n(c)_n} . \tag{5}$$

Here $R_n(z)$ is the remainder which we discuss later.

For the polynomials $B_n(z)$ and $A_n(z)$, we have

$$B_0(z) = 1,\ B_1(z) = 1 + \frac{(a+1)z}{2c},\ B_2(z) = 1 + \frac{(a+2)z}{2(c+1)} + \frac{(a+1)_2 z^2}{12(c)_2},$$

$$B_3(z) = 1 + \frac{(a+3)z}{2(c+2)} + \frac{(a+2)_2 z^2}{10(c+1)_2} + \frac{(a+1)_3 z^3}{120(c)_3},$$

$$A_0(z) = 1,\ A_1(z) = B_1(z) - \frac{az}{c},\ A_2(z) = B_2(z) - \frac{az}{c}\left[1 + \frac{(a+2)z}{2(c+1)}\right]$$

$$+ \frac{(a)_2 z^2}{2(c)_2},\ A_3(z) = B_3(z) - \frac{az}{c}\left[1 + \frac{(a+3)z}{2(c+2)} + \frac{(a+2)_2 z^2}{10(c+1)_2}\right]$$

$$+ \frac{(a)_2 z^2}{2(c)_2}\left[1 + \frac{(a+3)z}{2(c+2)}\right] - \frac{(a)_3 z^3}{6(c)_3} . \tag{6}$$

Both $A_n(z)$ and $B_n(z)$ satisfy the same recurrence formula

$$B_n(z) = (1 + F_1 z)B_{n-1}(z) + (E + F_2 z)zB_{n-2}(z) + F_3 z^3 B_{n-3}(z),$$

$$n \ge 3,\ F_1 = \frac{(n-a-2)}{2(2n-3)(n+c-1)},\ F_2 = \frac{(n+a)(n+a-1)}{4(2n-1)(2n-3)(n+c-2)_2},$$

$$F_3 = -\frac{(n+a-2)_2(n-a-2)}{8(2n-3)^2(2n-5)(n+c-3)_3},\ E = -\frac{(n+a-1)(n-c-1)}{2(2n-3)(n+c-2)_2} . \tag{7}$$

The recurrence formula is stable in the forward direction. We write the error in the form

$$R_n(z) = S_n(z)/B_n{}^*(z),\ B_n{}^*(z) = z^{-n}L_n{}^{-1}B_n(z), \tag{8}$$

and first consider the structure of and asymptotic forms for $S_n(z)$. A closed form representation for $S_n(z)$ can be derived after the manner of discussion given by Fields (4-6). We have

$$S_n(z) = F(z)M_n(z) + H(z)G_n(z), \tag{9}$$

where

$$F(z) = -\frac{az\Gamma(c)}{\Gamma(2-c)} e^{-z} {}_1F_1(1-a;2-c;z), \; H(z) = \frac{\Gamma(1-c)}{\Gamma(-a)} E(z), \tag{10}$$

$$M_n(z) = \frac{\Gamma(n+1-c)}{\Gamma(n+1+c)} \; {}_2F_2\left(\begin{matrix} c,\; c-a \\ c-n, n+1+c \end{matrix}\middle| z\right), \tag{11}$$

$$G_n(z) = \frac{(-)^{n+1} n!\Gamma(n+1-a)z^{n+1}}{\Gamma(n+2-c)(2n+1)!} \; {}_2F_2\left(\begin{matrix} n+1,\; n+1-a \\ 2n+2, n+2-c \end{matrix}\middle| z\right). \tag{12}$$

Asymptotic forms for $M_n(z)$ and $G_n(z)$ for n large with a, c and z fixed follow from the work of Luke (2,3). Thus

$$M_n(z) = \frac{\Gamma(n+1-c)}{\Gamma(n+1+c)}\left[{}_2F_2^r\left(\begin{matrix} c,\; c-a \\ c-n, n+c+1 \end{matrix}\middle| -z\right) + O(n^{-2r-2})\right], \tag{13}$$

$$\frac{\Gamma(n+1-c)}{\Gamma(n+1+c)} = (n+\tfrac{1}{2})^{-2c}[1+O(n^{-2})]. \tag{14}$$

$$G_n(z) = \frac{(-)^{n+1} n!\Gamma(n+1-a)z^{n+1}}{\Gamma(n+2-c)(2n+1)!}\left[\exp\left\{\frac{(n+1-a)z}{2(n+2-c)}\right\}\right][1+vz^2+O(n^{-2})$$

$$v = \frac{(n+1-a)}{8(2n+3)(n+3-c)(n+2-c)^2}[n^2+n(a-3c+6)+(a-5c+ac+7)],$$

$$\frac{n!\Gamma(n+1-a)}{\Gamma(n+2-c)(2n+1)!} = \frac{n!n^{c-a-1}}{(2n+1)!}[1+O(n^{-1})]. \tag{15}$$

If c is not a positive integer, then

$$S_n(z) = \frac{F(z)\Gamma(n+1-c)}{\Gamma(n+1+c)}\left[{}_2F_2^r\left(\begin{matrix}c,\ c-a\\ c-n, n+1+c\end{matrix}\middle| z\right) + O(n^{-2r-2})\right]. \quad (16)$$

If a is not a positive integer, but c is a positive integer, then

$$S_n(z) = \frac{(\cdot)^n\Gamma(a+1)\Gamma(n+1-c)}{\Gamma(a+1-c)\Gamma(n+1+c)} E(z)\left[{}_2F_2^r\left(\begin{matrix}c,\ c-a\\ c-n, n+1+c\end{matrix}\middle| z\right) + O(n^{-2r-2})\right], \quad (17)$$

where $E(z)$ is given by (1). If both c and a are positive integers, $c > a$, then

$$S_n(z) = -\frac{az^{1-c}\Gamma(c-1)e^{-z}}{\Gamma(c-a)}\ {}_1F_1^{a-1}\ (1-a;2-c;z)G_n(z). \quad (18)$$

With a, c and z fixed, it can be shown from the work of Luke (2,3) that

$$B_n{}^*(z) = \frac{(c)_n(2n)!}{(a+1)_n z^n n!}\left[\exp\left\{\frac{(n+a)z}{2(n+c-1)}\right\}\right]\ [1 - uz^2 + O(n^{-2})] ,$$

$$u = \frac{(n+a)}{8(2n-1)(n+c-1)(n+c-2)^2}\ [n^2+n\{(3-2c)(2-a-c)+2(c-a-1)\}+ac+2-2c]. \quad (19)$$

If the numbers c and a are arbitrary except as previously noted with the further provision that if these numbers are positive integers, they are not so simutaneously, then the forms for the error readily follow from (8), (9) and (16) or (17) as appropriate. In these situations, we have

$$R_n(z) = \frac{W\Gamma(n+1-c)\Gamma(n+a+1)n!z^n}{\Gamma(n+1+c)\Gamma(n+c)(2n)!}\ [1 + O(n^{-1})] \quad (20)$$

where W is free of n except that it might contain the factor $(-)^n$. Clearly

$$\lim_{n\to\infty} R_n(z) = 0 \tag{21}$$

and

$$\left|R_{n+1}(z)/R_n(z)\right| = \left|\frac{(n+1-c)(n+a+1)z}{2(n+c)(n+c+1)(2n+1)}\right| [1 + O(n^{-1})]. \tag{22}$$

If both c and a are positive integers with $c > a$, then from (8), (9), (15) and (18) we have

$$R_n(z) = \frac{(-)^n z^{2n+2-c} e^{-z} \Gamma(c)\Gamma(c-1)(n!)^2 \Gamma(n+1-a)\Gamma(n+1+a)}{\Gamma(a)\Gamma(c-a)\Gamma(n+c)\Gamma(n+2-c)(2n)!(2n+1)!} \, {}_1F_1^{a-1}\left(\begin{matrix}1-a\\2-c\end{matrix}\middle| z\right)$$

$$\times \exp\left\{\frac{z(c-a-1)(2n+1)}{2(n+c-1)(n+2-c)}\right\} [1 + (u+v)z^2 + O(n^{-2})]. \tag{23}$$

Again

$$\lim_{n\to\infty} R_n(z) = 0 \ , \tag{24}$$

and

$$R_{n+1}(z)/R_n(z) = -\frac{z^2(n+1-a)(n+1+a)}{4(2n+1)(2n+3)(n+c)(n+2-c)} [1 + O(n^{-1})]. \tag{25}$$

It is of interest to compare the error $R_n(z)$ where $a = 1$ with the corresponding error, call it $R_{n,p}(z)$, for the Pade approximations in the next chapter. If c is not a positive integer,

$$\frac{R_{n,p}(z)}{R_n(z)} = \frac{(-)^{n+1} z^n e^{z/2} (\pi/n)^{3/2}}{\Gamma(c)\Gamma(c-1)(\sin \pi c) n! \, 2^{2n+2c-1}} [1 + O(n^{-1})] \ , \tag{26}$$

whence the Padé approximation is superior. Now consider (23).

If z is small,

$$R_n(z) = O(z^{2n+2-c}). \tag{27}$$

This would be the situation for a Padé approximation if the numerator and denominator polynomials were of degree $n+1-c$ and n respectively. Thus our rational approximations under the conditions leading to (23), though not of the Padé class, are very much akin to this class. Indeed, when $a = 1$ and c is a positive integer, $c > 1$, we find that

$$\frac{R_{n,p}(z)}{R_n(z)} = -(z/4n)^{c-1}[1 + O(n^{-1})] , \tag{28}$$

and again the Padé approximation is superior.

Numerical Examples

Let $a = 2/3$, $c = 4/3$, $z = 3/4$ and $n = 5$. Then from (8), (9) and (16) without order terms,

$$B_n(z) = 1.47509,\ B_n^*(z) = 1.04541(5),\ R_n(z) = -0.23921(-7).$$

The true values of $B_n(z)$ and $R_n(z)$ are 1.47779 and -0.23914(-7), respectively.

Suppose $a = 1$, $c = 2$ and $z = 5/4$. If $n = 5$, then from (23) without order terms, $R_n(z) = -0.2762(-9)$ which agrees with the true value.

3. Rational Approximations for ${}_1F_1(1;c;-z)$

Let

$$E(z) = {}_1F_1(1;c;-z) \tag{1}$$

which is a form of the incomplete gamma function. See Luke (2,3). The rational approximations in this section lie on the main diagonal of the Padé table. If $c = 1$, we get approximations for the exponential function, see later comments.

We write

$$E(z) = \{A_n(z)/B_n(z)\} + R_n(z) , \tag{2}$$

$$B_n(z) = L_n z^n \, {}_2F_0(-n,n+c;-\tfrac{1}{z}) \tag{3}$$

$$A_n(z) = L_n z^n \sum_{k=0}^{n} \frac{(-n)_k(n+c)_k}{(c)_k k!} \, {}_3F_1\left(\begin{matrix}-n+k,n+c+k,1\\ 1+k\end{matrix}\middle| -\frac{1}{z}\right) \tag{4}$$

$$L_n = \frac{\Gamma(n+c)}{\Gamma(2n+c)} \tag{5}$$

where $R_n(z)$ is the remainder which is discussed later.

For the polynomials $A_n(z)$ and $B_n(z)$, we have

$$B_0(z) = 1,\ B_1(z) = 1 + \frac{z}{c+1} ,\ B_2(z) = 1 + \frac{2z}{c+3} + \frac{z^2}{(c+2)_2} ,$$

$$B_3(z) = 1 + \frac{3z}{c+5} + \frac{3z^2}{(c+4)_2} + \frac{z^3}{(c+3)_3} ,$$

$$A_0(z) = 1,\ A_1(z) = B_1(z) - \frac{z}{c} ,\ A_2(z) = B_2(z) - \frac{z}{c}\left(1 + \frac{2z}{c+3}\right) + \frac{z^2}{(c)_2} ,$$

$$A_3(z) = B_3(z) - \frac{z}{c}\left(1 + \frac{3z}{c+5} + \frac{3z^2}{(c+4)_2}\right) + \frac{z^2}{(c)_2}\left(1 + \frac{3z}{c+5}\right) - \frac{z^3}{(c)_3} . \tag{6}$$

Both $A_n(z)$ and $B_n(z)$ satisfy the same recursion formula

$$B_n(z) = (1 + F_1 z)B_{n-1}(z) + F_2 z^2 B_{n-2}(z), \; n \geq 2 ,$$

$$F_1 = \frac{(c-1)}{(2n+c-1)(2n+c-3)} , \; F_2 = \frac{(n-1)(n+c-2)}{(2n+c-2)(2n+c-3)^2(2n+c-4)} . \tag{7}$$

The recursion formula is stable in the forward direction.

For the remainder, we have

$$R_n(z) = \frac{(-)^{n+1}\pi\Gamma(c)n!\Gamma(n+c)z^{2n+1}\{\exp[-z+z(z+4c-4)/4(2n+c)]\}[1+O(n^{-3})]}{2^{4n+2c-2}(2n+c)[\Gamma(n+c/2)\Gamma(n+(c+1)/2)]^2}, \tag{8}$$

or

$$R_n(z) = \frac{(-)^{n+1}\pi\Gamma(c)n^{1-c}z^{2n+1}\{\exp[-z+z(z+4c-4)/4(2n+c)]\}[1+O(n^{-1})]}{2^{4n+2c-1}(n!)^2}. \tag{9}$$

It follows that for z and c fixed,

$$\lim_{n\to\infty} R_n(z) = 0. \tag{10}$$

To facilitate computation of a priori error estimates, we have

$$\frac{R_{n+1}(z)}{R_n(z)} = -\frac{(n+1)(n+c)z^2}{(2n+c)(2n+c+1)^2(2n+c+2)} \exp\left[\frac{-z(z+4c-4)}{2(2n+c)(2n+c+2)}\right][1+O(n^{-3})]$$

$$= -\frac{z^2}{4(2n+c+1)^2}[1 + O(n^{-2})] , \tag{11}$$

which is a measure of the rate of convergence. With $R_n(z,c)$ in

place of $R_n(z)$,

$$\frac{R_n(z,c+h)}{R_n(c)} = \frac{\Gamma(c+h)}{\Gamma(c)} \left(\frac{n+c}{(2n+c)(2n+c+1)}\right)^h \exp\left[\frac{-zh(8n+4-z)}{4(2n+c)(2n+c+h)}\right] [1+O(n^{-1})]. \tag{12}$$

The latter shows that for a given z, c and n, the error changes but slightly for small values of h, that is, for small changes in c.

Next we consider some results for the exponential function - the case $c = 1$. We have

$$e^{-z} = \{G_n(-z)/G_n(z)\} + S_n(z) , \tag{13}$$

where

$$G_n(z) = \frac{(2n)!}{n!} B_n(z) , \tag{14}$$

$$G_n(z) = z^n \; {}_2F_0(-n,n+1;-1/z) = \frac{(2n)!}{n!} \; {}_1F_1^n(-n;-2n;z) , \tag{15}$$

and $S_n(z) = R_n(z)$ with $c = 1$. It is convenient to write

$$G_n(z) = M_n(z^2) + zN_n(z^2). \tag{16}$$

Then by computing $M_n(z^2)$ and $N_n(z^2)$, evaluation of the main diagonal Padé approximation of order n only necessitates the evaluation of essentially $(n+1)$ terms. The polynomials $G_n(z)$, $M_n(z^2)$ and $N_n(z^2)$ satisfy the same recurrence formula,

$$G_{n+1}(z) = 2(2n+1)G_n(z) + z^2G_{n-1}(z) , \tag{17}$$

$$G_0(z) = 1,\ G_1(z) = z+2,\ G_2(z) = z^2 + 6z + 12\ ,$$

$$G_3(z) = z^3 + 12\ z^2 + 60\ z + 120,$$

$$G_4(z) = z^4 + 20\ z^3 + 180\ z^2 + 840\ z + 1680. \tag{18}$$

We also have the explicit representations,

$$M_n(z^2) = \frac{(2n)!}{n!}\ {}_2F_3^{[n/2]}\left(\begin{matrix}-\tfrac{1}{2}n,\ \tfrac{1}{2} - \tfrac{1}{2}n \\ -n,\ \tfrac{1}{2} - n,\ \tfrac{1}{2}\end{matrix}\middle| z^2/4\right)\ , \tag{19}$$

$$N_n(z^2) = \frac{2(2n)!}{n!}\ {}_2F_3^{[\frac{1}{2}n-\frac{1}{2}]}\left(\begin{matrix}\tfrac{1}{2} - \tfrac{1}{2}n,\ 1 - \tfrac{1}{2}n \\ 1 - n,\ \tfrac{1}{2} - n,\ 3/2\end{matrix}\middle| z^2/4\right)\ , \tag{20}$$

where $[y]$ stands for the largest integer in y.

As previously noted, forms for the error $S_n(z)$ follow from (8) - (11) with $c = 1$. In closed form,

$$S_n(z) = \frac{(-)^{n+1}\pi e^{-z} I_{n+\frac{1}{2}}(z/2)}{K_{n+\frac{1}{2}}(z/2)}\ , \tag{21}$$

and we have the asymptotic representation

$$S_n(z) = (-)^{n+1}e^{-z}[\exp(2\nu\zeta + 2U/\nu)][1 + O(\nu^{-3})] \tag{22}$$

uniformly in z, $z \neq 0$, $|\arg z| < \pi/2$, where

$$\nu = n + \tfrac{1}{2},\ z = 2\nu x,\ \zeta = u^{-1} + \ln\left(\frac{ux}{1+u}\right)\ ,\ u = (1 + x^2)^{-\frac{1}{2}},$$

$$U = (3u - 5u^3)/24. \tag{23}$$

In particular, for z large, we have

$$S_n(z) = (-)^{n+1} \exp\left[-\frac{\nu}{x}\left(1 - \frac{1}{12\,x^2} + O(x^{-3})\right)\right] \exp\left[\frac{2U}{\nu}\right][1+O(\nu^{-3})]. \tag{24}$$

In illustration, let $n = 4$, $z = 9$ whence $\nu = 9/2$. Neglecting order terms, from (22) we get $S_4(z) = -0.01474$ whereas the true value is -0.01503. Uniform asymptotic representations for the first and second subdiagonal Padé approximations for e^{-z} are also given in (7). For further comments on Padé approximations for (1) and other remarks on the exponential function, see (2,3,7, 8,9). In connection with reference (7) Dr. M.G. de Bruin, Universiteit van Amsterdam, Instituut voor Propedeutische Wiskunde, Roetersstraat 15, Amsterdam, Netherlands has kindly informed me that in quoting the results of H. van Rossum, I overlooked the restriction $\mu \le \nu+1$. Consequently the general results of reference (7) are not valid for the lower part of the Padé table unless $c = 0$. The case $c = 0$ is for the exponential function. Also the concept of 'normal' employed by H. van Rossum is different from that usually employed for the Padé table.

Numerical Examples

Let $c = 2$. Then

$${}_1F_1(1;2;-z) = (1 - e^{-z})/z.$$

We take $z = \frac{1}{2}$. If $V_1(n)$ is the right hand side of (8) with

$O(n^{-3})$ neglected, then

$$V_1(n) = \frac{(-)^{n+1}\exp[-(16n+7)/32(n+1)]}{2^{6n+2}[(3/2)_n]^2} .$$

Values of $V_1(n)$ and the true values of $V(n)$ for $n = 0(1)3$ are recorded in the following table.

n	$(-)^{n+1}V_1(n)$	$(-)^{n+1}V_1(n)$
0	0.201	0.213
1	0.121(-2)	0.122(-2)
2	0.289(-5)	0.290(-5)
3	0.360(-8)	0.361(-8)

Thus the values of $V_1(n)$ are remarkably accurate even for small values of n. If $V_2(n)$ is the right hand side of (9) with $O(n^{-1})$ neglected, then

$$V_2(n) = \frac{(-)^{n+1}\exp[-(16n+7)/32(n+1)]}{2^{6n+4}n(n!)^2} .$$

Values of $V_2(n)$ and the true values of $V(n)$ for $n = 3,4,5$ are recorded in the following table.

n	$(-)^{n+1}V_2(n)$	$(-)^{n+1}V(n)$
3	0.451(-8)	0.361(-8)
4	0.326(-11)	0.274(-11)
5	0.161(-14)	0.140(-14)

The above is the sample problem treated in the programs which follow.

References

1. Y.L. Luke, "Algorithms for Rational Approximations for a Confluent Hypergeometric Function," to be published in Utilitas Mathematica.

2. Y.L. Luke, "The Special Functions and Their Approximations," Vols. 1,2, Academic Press, New York, 1969.

3. Y.L. Luke, "Mathematical Functions and Their Approximations," Academic Press, New York, 1975.

4. J.L. Fields, "A Linear Scheme for Rational Approximations," J. Approx. Theory 6 (1972), 161-175.

5. J.L. Fields, "Uniform Asymptotic Expansions of Certain Classes of Meijer G-Functions for a Large Parameter," SIAM J. Math. Anal. 4 (1973), 482-507.

6. J.L. Fields, written communication, 1976.

7. Y.L. Luke, "On the Error in the Padé Approximants for a Form of the Incomplete Gamma Function Including the Exponential Function," SIAM J. Math. Anal. 6 (1975), 829-839.

8. Y.L. Luke, "Evaluation of the Gamma Function by Means of Padé Approximations," SIAM J. Math. Anal. 1 (1970), 266-281.

9. Y.L. Luke, "Chebyshev Expansions and Rational Approximations for Some Special Functions and Analytic Continuation Formulas for These Special Functions," to be published in the Journal of Computation and Applied Mathematics.

```
C     .................................................................
C     .        THIS MAINLINE PROGRAM ILLUSTRATES THE SYNTAX OF THE SUB-  .
C     .  ROUTINE 'R1F1' FOR GENERATING VALUES OF THE NUMERATOR AND      .
C     .  DENOMINATOR POLYNOMIALS IN THE RATIONAL APPROXIMATION OF       .
C     .  1F1( AP ; CP ; -Z ) .                                          .
C     .................................................................
      IMPLICIT REAL*16(A-H,O-Z)
      DIMENSION A(26),B(26),R(26),D(26),E(26)
      DATA ZERO/0.Q0/
 10   READ(5,1,END=999) N,M
      READ(5,2) AP,CP
      N1=N+1
      D(N1)=ZERO
      E(N1)=ZERO
      DO 100 I=1,M
         READ(5,2) Z
C-----------------------------------------------------------------------
         CALL R1F1(AP,CP,Z,A,B,N)
C-----------------------------------------------------------------------
C     IN THE ABOVE :
C
C  AP           IS THE NUMERATOR PARAMETER OF THE 1F1
C  CP           IS THE DENOMINATOR PARAMETER OF THE 1F1
C  Z            IS THE VALUE OF THE ARGUMENT
C  A AND B      WILL CONTAIN THE VALUES OF THE NUMERATOR AND DENOMINATOR
C                  POLYNOMIALS, RESPECTIVELY, FOR ALL DEGREES FROM 0 TO
C                  N INCLUSIVE
C  N            IS THE MAXIMUM DEGREE FOR WHICH VALUES OF THE POLYNOMIALS
C                  ARE TO BE CALCULATED
C
C     NOTE : VALUES OF THE K-TH DEGREE POLYNOMIALS WILL BE PLACED IN
C A(K+1) AND B(K+1) RESPECTIVELY.
C-----------------------------------------------------------------------
         R(N1)=A(N1)/B(N1)
         DO 50 J=1,N
            J1=N1-J
            R(J1)=A(J1)/B(J1)
            D(J1)=R(J1+1)-R(J1)
 50         E(J1)=R(N1)-R(J1)
         WRITE(6,3) N,AP,CP,Z
         DO 60 J=1,N1
            J1=J-1
 60         WRITE(6,4) J1,A(J),B(J)
         WRITE(6,5)
         DO 70 J=1,N1
            J1=J-1
 70         WRITE(6,6) J1,R(J),D(J),E(J)
 100     CONTINUE
      GOTO 10
 999  STOP
 1    FORMAT(2I2)
 2    FORMAT(Q39.32)
 3    FORMAT('1','VALUES OF THE POLYNOMIALS IN THE RATIONAL APPROXIMATIO
     :N OF 1F1(AP;CP;-Z)'//' ','N = ',I2,T20,'AP = ',Q39.32//' ',T20,'CP
     : = ',Q39.32//' ','     Z = ',Q39.32//' ',' I',T24,'A(I)',T65,
```

```
     :'B(I)'/)
4     FORMAT(' ',I2,2X,Q39.32,2X,Q39.32)
5     FORMAT('0','VALUES OF THE APPROXIMATION, 1ST DIFFERENCES AND APPRO
     :XIMATE ERRORS'//' ',' I',T12,'I-TH APPROXIMATION -- F(I)',T47,
     :'1ST DIFF''S.',T60,'F(N)-F(I)'/)
6     FORMAT(' ',I2,2X,Q39.32,2X,Q10.3,2X,Q10.3)
      END
```

```
      SUBROUTINE R1F1(AP,CP,Z,A,B,N)
C     *****************************************************************
C     *       THIS SUBROUTINE RETURNS VALUES A(I) AND B(I), I=1,...,N+1 *
C     *  OF THE NUMERATOR AND DENOMINATOR POLYNOMIALS IN THE RATIONAL   *
C     *  APPROXIMATION OF 1F1( AP ; CP ; -Z ) .                         *
C     *                                                                 *
C     *       NO OTHER SUBROUTINES ARE CALLED BY THIS ONE.              *
C     *****************************************************************
      IMPLICIT REAL*16(A-H,O-Z)
      DIMENSION A(1),B(1)
      DATA ZERO/0.Q0/,ONE/1.Q0/,TWO/2.Q0/,THREE/3.Q0/
C
C  INITIALIZATION :
C
      CT1=AP*Z/CP
      XN3=ZERO
      XN1=TWO
      Z2=Z/TWO
      CT2=Z2/(ONE+CP)
      XN2=ONE
      A(1)=ONE
      B(1)=ONE
      B(2)=ONE+(ONE+AP)*Z2/CP
      A(2)=B(2)-CT1
      B(3)=ONE+(TWO+B(2))*(TWO+AP)/THREE*CT2
      A(3)=B(3)-(ONE+CT2)*CT1
      CT1=THREE
      XN0=THREE
C
C  FOR I=3,...,N , THE VALUES A(I+1) AND B(I+1) ARE CALCULATED
C  USING THE RECURRENCE RELATIONS BELOW.
C
      DO 100 I=3,N
C
C  CALCULATION OF THE MULTIPLIERS FOR THE RECURSION
C
         CT2=Z2/CT1/(CP+XN1)
         G1=ONE+CT2*(XN2-AP)
         CT2=CT2*(AP+XN1)/(CP+XN2)
         G2=CT2*((CP-XN1)+(AP+XN0)/(CT1+TWO)*Z2)
         G3=CT2*Z2*Z2/CT1/(CT1-TWO)*(AP+XN2)/(CP+XN3)*(AP-XN2)
C-----------------------------------------------------------------------
C  THE RECURRENCE RELATIONS FOR A(I+1) AND B(I+1) ARE AS FOLLOWS
C-----------------------------------------------------------------------
C
         B(I+1)=G1*B(I)+G2*B(I-1)+G3*B(I-2)
         A(I+1)=G1*A(I)+G2*A(I-1)+G3*A(I-2)
C
         XN3=XN2
         XN2=XN1
         XN1=XN0
         XN0=XN0+ONE
  100    CT1=CT1+TWO
      RETURN
      END
```

VALUES OF THE POLYNOMIALS IN THE RATIONAL APPROXIMATION OF 1F1(AP;CP;-Z)

N = 12 AP = 0.66666666666666666666666666666667Q+00

CP = 0.13333333333333333333333333333333Q+01

Z = 0.75000000000000000000000000000000Q+00

I	A(I)	B(I)
0	0.10000000000000000000000000000000Q+01	0.10000000000000000000000000000000Q+01
1	0.10937500000000000000000000000000Q+01	0.14687500000000000000000000000000Q+01
2	0.10602678571428571428571428571429Q+01	0.14955357142857142857142857142857Q+01
3	0.10543247767857142857142857142857Q+01	0.14887388392857142857142857142857Q+01
4	0.10497612251030219780219780219780Q+01	0.14823585594093406593406593406593Q+01
5	0.10465263492458469265109890109890Q+01	0.14777924841576880151098901098901Q+01
6	0.10441792686314379493861401756138Q+01	0.14744782368037043651250065723750Q+01
7	0.10424143361305815529729144383404Q+01	0.14719859890943789052857237191406Q+01
8	0.10410437543466518752500583298280Q+01	0.14700506001719704398384818306693Q+01
9	0.10399505127119027879130077349967Q+01	0.14685068413106010306638025870989Q+01
10	0.10390589966739233429961726058947Q+01	0.14672479377523064477628289447878Q+01
11	0.10383184973928179838167173810509Q+01	0.14662022838995543814800803421874Q+01
12	0.10376938567505824346979621559842Q+01	0.14653202332199585623597053053488Q+01

VALUES OF THE APPROXIMATION, 1ST DIFFERENCES AND APPROXIMATE ERRORS

I	I-TH APPROXIMATION -- F(I)	1ST DIFF'S.	F(N)-F(I)
0	0.10000000000000000000000000000000Q+01	-0.255Q+00	-0.292Q+00
1	0.74468085106382978234042553191480Q+00	-0.357Q-01	-0.365Q-01
2	0.70895522388059701492537313432835Q+00	-0.755Q-03	-0.787Q-03
3	0.70819995352010255564468367431085Q+00	-0.304Q-04	-0.313Q-04
4	0.70816957101209639178728188451489Q+00	-0.893Q-06	-0.917Q-06
5	0.70816867758151164976432828063979Q+00	-0.234Q-07	-0.239Q-07
6	0.70816865421829102586102754734846Q+00	-0.540Q-09	-0.551Q-09
7	0.70816865367850004085947369770368Q+00	-0.112Q-10	-0.114Q-10
8	0.70816865366734165504171086103969Q+00	-0.208Q-12	-0.212Q-12
9	0.70816865366713322256848696186163Q+00	-0.355Q-14	-0.360Q-14
10	0.70816865366712967389837210139300Q+00	-0.555Q-16	-0.563Q-16
11	0.70816865366712961841950792237845Q+00	-0.802Q-18	-0.802Q-18
12	0.70816865366712961761794786929662Q+00	0.0	0.0

```
C     .............................................................
C     .        THIS MAINLINE PROGRAM ILLUSTRATES THE SYNTAX OF THE SUB-  .
C     .  ROUTINE 'C1F1' WHEN USED TO GENERATE COEFFICIENTS IN THE POLY-  .
C     .  NOMIALS FOR THE RATIONAL APPROXIMATION OF 1F1(AP;CP;-Z).        .
C     .............................................................
      IMPLICIT REAL*16(A-H,O-Z)
      DIMENSION CA(26),CB(26),NO(25)
 10   READ(5,2,END=999) AP,CP
      WRITE(6,3) AP,CP
      READ(5,1) M,(NO(J),J=1,M)
      DO 100 I=1,M
         N=NO(I)
C-----------------------------------------------------------------
         CALL C1F1(AP,CP,CA,CB,N)
C-----------------------------------------------------------------
C        IN THE ABOVE:
C
C  AP            IS THE NUMERATOR PARAMETER OF THE 1F1
C  CP            IS THE DENOMINATOR PARAMETER OF THE 1F1
C  N             IS THE DEGREE OF THE POLYNOMIALS IN THE RATIONAL
C                    APPROXIMATION
C  CA AND CB     WILL CONTAIN THE COEFFICIENTS IN THE NUMERATOR AND
C                    DENOMINATOR POLYNOMIALS RESPECTIVELY
C
C        NOTE : THE COEFFICIENTS OF THE K-TH POWER OF Z WILL BE PLACED
C IN CA(K+1) OR CB(K+1) AS APPROPRIATE
C-----------------------------------------------------------------
         N1=N+1
 100     WRITE(6,4) N,(CA(J),CB(J),J=1,N1)
      GOTO 10
 999  STOP
 1    FORMAT(26I2)
 2    FORMAT(Q39.32)
 3    FORMAT('1','COEFFICIENTS FOR THE RATIONAL APPROXIMATION OF 1F1( AP
     : ; CP ; -Z )'//' ',T20,'AP = ',Q39.32/' ',T20,'CP = ',Q39.32/)
 4    FORMAT(' ','N = ',I2,T18,'CA(I)',T58,'CB(I)'//
     :26(1X,Q39.32,2X,Q39.32/)/)
      END
```

```
      SUBROUTINE C1F1(AP,CP,A,B,N)
C     ************************************************************************
C     *        THIS SUBROUTINE RETURNS COEFFICIENTS A(I) AND B(I) ,           *
C     *  I = 1,2,...,N+1 , OF THE NUMERATOR AND DENOMINATOR POLYNOMIALS       *
C     *  RESPECTIVELY, IN THE RATIONAL APPROXIMATION OF ORDER  N  FOR         *
C     *  1F1( AP ; CP ; -Z ).                                                 *
C     *                                                                       *
C     *        NO OTHER SUBROUTINES ARE CALLED BY THIS ONE.                   *
C     ************************************************************************
      IMPLICIT REAL*16(A-H,O-Z)
      DIMENSION A(1),B(1)
      DATA ONE/1.Q0/,ZERO/0.Q0/
C
C  INITIALIZATION :
C
      XN=N
      XN1I=XN
      CP1=CP-ONE
      B(1)=ONE
      A(1)=ONE
      XI=ONE
      XIJ=ZERO
      DO 100 I=1,N
         I1=I+1
C---------------------------------------------------------------------------
C  FOR I = 1,2,...,N , B(I+1) IS COMPUTED AS FOLLOWS
C---------------------------------------------------------------------------
C
         B(I1)=(AP+XN1I)/(CP1+XN1I)*XN1I/(XN+XN1I)*B(I)/XI
C
         A(I1)=ONE
         DO 50 J=1,I
C---------------------------------------------------------------------------
C  TO CALCULATE A(I+1), WE EMPLOY B(J) , J = 1,2,...,I+1 AS FOLLOWS
C---------------------------------------------------------------------------
C
            A(I1)=B(J+1)-(AP+XIJ)/(CP+XIJ)*A(I1)/(ONE+XIJ)
C
 50         XIJ=XIJ-ONE
         XIJ=XI
         XN1I=XN-XI
 100     XI=XI+ONE
      RETURN
      END
```

COEFFICIENTS FOR THE RATIONAL APPROXIMATION OF 1F1(AP ; CP ; -Z)

AP = 0.66666666666666666666666666666667Q+00
CP = 0.13333333333333333333333333333333Q+01

N = 3	CA(I)	CB(I)
	0.10000000000000000000000000000000Q+01	0.10000000000000000000000000000000Q+01
	0.49999999999999999999999999999991Q-01	0.55000000000000000000000000000001Q+00
	0.29285714285714285714285714285714Q-01	0.12571428571428571428571428571429Q+00
	0.83333333333333333333333333333331Q-03	0.13095238095238095238095238095239Q-01

N = 4	CA(I)	CB(I)
	0.10000000000000000000000000000000Q+01	0.10000000000000000000000000000000Q+01
	0.38461538461538461538461538461527Q-01	0.53846153846153846153846153846154Q+00
	0.36263736263736263736263736263737Q-01	0.12692307692307692307692307692308Q+00
	0.11904761904761904761904761904760Q-02	0.16117216117216117216117216117217Q-01
	0.45787545787545787545787545787567Q-04	0.10073260073260073260073260073261Q-02

N = 5	CA(I)	CB(I)
	0.10000000000000000000000000000000Q+01	0.10000000000000000000000000000000Q+01
	0.31249999999999999999999999999989Q-01	0.53125000000000000000000000000000Q+00
	0.40083180708180708180708180708181Q-01	0.12713675213675213675213675213675Q+00
	0.11599511599511599511599511599511Q-02	0.17481303418803418803418803418804Q-01
	0.16502594627594627594627594627593Q-03	0.14270451770451770451770451770452Q-02
	0.20667989417989417989417989418095Q-05	0.59460215710215710215710215710219Q-04

N = 6	CA(I)	CB(I)
	0.10000000000000000000000000000000Q+01	0.10000000000000000000000000000000Q+01
	0.26315789473684210526315789473671Q-01	0.52631578947368421052631578947369Q+00
	0.42506835269993164730006835269994Q-01	0.12709330143540669856459330143541Q+00
	0.10685586343481080323185586343478Q-02	0.18249294565084038768249294565084Q-01
	0.25404055009318167212904055009322Q-03	0.16728520017993702204228520017994Q-02
	0.48014850646429593798014850646370Q-05	0.95591542959964012595591542959967Q-04
	0.79111921217184375079111921220009Q-07	0.28449863976179765653449863976181Q-05

```
C     ...................................................................
C     .        THIS MAINLINE PROGRAM ILLUSTRATES THE SYNTAX OF THE SUB-  .
C     .  ROUTINE 'R1F1P' FOR GENERATING VALUES OF THE NUMERATOR AND      .
C     .  DENOMINATOR POLYNOMIALS IN THE PADE APPROXIMATION OF            .
C     .  1F1( 1 ; CP ; -Z ) .                                            .
C     ...................................................................
      IMPLICIT REAL*16(A-H,O-Z)
      DIMENSION A(26),B(26),R(26),D(26),E(26)
      DATA ZERO/0.Q0/
 10   READ(5,1,END=999) N,M
      READ(5,2) CP
      N1=N+1
      D(N1)=ZERO
      E(N1)=ZERO
      DO 100 I=1,M
         READ(5,2) Z
C----------------------------------------------------------------------
         CALL R1F1P(CP,Z,A,B,N)
C----------------------------------------------------------------------
C     IN THE ABOVE :
C
C  CP            IS THE DENOMINATOR PARAMETER OF THE 1F1
C  Z             IS THE VALUE OF THE ARGUMENT
C  A AND B       WILL CONTAIN THE VALUES OF THE NUMERATOR AND DENOMINATOR
C                   POLYNOMIALS, RESPECTIVELY, FOR ALL DEGREES FROM 0 TO
C                   N INCLUSIVE
C  N             IS THE MAXIMUM DEGREE FOR WHICH VALUES OF THE POLYNOMIALS
C                   ARE TO BE CALCULATED
C
C     NOTE : VALUES OF THE K-TH DEGREE POLYNOMIALS WILL BE PLACED IN
C A(K+1) AND B(K+1) RESPECTIVELY.
C----------------------------------------------------------------------
         R(N1)=A(N1)/B(N1)
         DO 50 J=1,N
            J1=N1-J
            R(J1)=A(J1)/B(J1)
            D(J1)=R(J1+1)-R(J1)
 50         E(J1)=R(N1)-R(J1)
         WRITE(6,3) N,CP,Z
         DO 60 J=1,N1
            J1=J-1
 60         WRITE(6,4) J1,A(J),B(J)
         WRITE(6,5)
         DO 70 J=1,N1
            J1=J-1
 70         WRITE(6,6) J1,R(J),D(J),E(J)
 100     CONTINUE
      GOTO 10
 999  STOP
 1    FORMAT(2I2)
 2    FORMAT(Q39.32)
 3    FORMAT('1','VALUES OF THE POLYNOMIALS IN THE PADE APPROXIMATION OF
     : 1F1(1;CP;-Z)'//' ','N = ',I2,T20,'CP = ',Q39.32//' ','    Z = ',
     :Q39.32//' ',' I',T24,'A(I)',T65,'B(I)'/)
 4    FORMAT(' ',I2,2X,Q39.32,2X,Q39.32)
```

```
5     FORMAT('0','VALUES OF THE APPROXIMATION, 1ST DIFFERENCES AND APPRO
     :XIMATE ERRORS'//' ',' I',T12,'I-TH APPROXIMATION -- F(I)',
     :T47,'1ST DIFF''S.',T60,'F(N)-F(I)'/)
6     FORMAT(' ',I2,2X,Q39.32,2X,Q10.3,2X,Q10.3)
      END
```

```
      SUBROUTINE R1F1P(CP,Z,A,B,N)
C     ******************************************************************
C     *       THIS SUBROUTINE RETURNS VALUES A(I) AND B(I), I=1,...,N+1 *
C     *  OF THE NUMERATOR AND DENOMINATOR POLYNOMIALS IN THE PADE       *
C     *  APPROXIMATION OF 1F1( 1 ; CP ; -Z ) .                          *
C     *                                                                 *
C     *       NO OTHER SUBROUTINES ARE CALLED BY THIS ONE.              *
C     ******************************************************************
      IMPLICIT REAL*16(A-H,O-Z)
      DIMENSION A(1),B(1)
      DATA ONE/1.Q0/,TWO/2.Q0/
C
C  INITIALIZATION :
C
      XI1=ONE
      B(1)=ONE
      A(1)=ONE
      CT1=CP+ONE
      CP1=CP-ONE
      ZZ=Z*Z
      B(2)=ONE+Z/CT1
      A(2)=B(2)-Z/CP
C
C  FOR I=2,...,N , THE VALUES A(I+1) AND B(I+1) ARE CALCULATED
C  USING THE RECURRENCE RELATIONS BELOW.
C
      DO 100 I=2,N
C
C  CALCULATION OF THE MULTIPLIERS FOR THE RECURSION
C
         CT2=CT1*CT1
         G1=ONE+CP1/(CT2+CT1+CT1)*Z
         G2=XI1/(CT2-ONE)*(XI1+CP1)/CT2*ZZ
C-----------------------------------------------------------------------
C  THE RECURRENCE RELATIONS FOR A(I+1) AND B(I+1) ARE AS FOLLOWS
C-----------------------------------------------------------------------
C
         A(I+1)=G1*A(I)+G2*A(I-1)
         B(I+1)=G1*B(I)+G2*B(I-1)
C
         CT1=CT1+TWO
 100     XI1=XI1+ONE
      RETURN
      END
```

VALUES OF THE POLYNOMIALS IN THE PADE APPROXIMATION OF 1F1(1;CP;-Z)

N = 12 CP = 0.20000000000000000000000000000000Q+01

Z = 0.50000000000000000000000000000000Q+00

I	A(I)	B(I)
0	0.10000000000000000000000000000000Q+01	0.10000000000000000000000000000000Q+01
1	0.91666666666666666666666666666666Q+00	0.11666666666666666666666666666667Q+01
2	0.95416666666666666666666666666666Q+00	0.12125000000000000000000000000000Q+01
3	0.97008928571428571428571428571428Q+00	0.12327380952380952380952380952381Q+01
4	0.97900545634920634920634920634920Q+00	0.12440682870370370370370370370370Q+01
5	0.98469845403439153439153439153438Q+00	0.12513026470057720057720057720057Q+01
6	0.98864714025846838346838346838346Q+00	0.12563204283419127169127169127169Q+01
7	0.99154629655524837816504483171149Q+00	0.12600045226282205448872115538782Q+01
8	0.99376512918549386456861947058024Q+00	0.12628240976281747766122766122766Q+01
9	0.99551790005149933812496699493602Q+00	0.12650514260201726818204823880778Q+01
10	0.99693746696649013147801190978987Q+00	0.12668553364772797415581948528899Q+01
11	0.99811058146757188014005211141739Q+00	0.12683460682585984278952231472198Q+01
12	0.99909628932987904214020674225976Q+00	0.12695986536081802525845977971745Q+01

VALUES OF THE APPROXIMATION, 1ST DIFFERENCES AND APPROXIMATE ERRORS

I	I-TH APPROXIMATION -- F(I)	1ST DIFF'S.	F(N)-F(I)
0	0.10000000000000000000000000000000Q+01	-0.214Q+00	-0.213Q+00
1	0.78571428571428571428571428571429Q+00	0.123Q-02	0.122Q-02
2	0.78694158075601374570446735395189Q+00	-0.290Q-05	-0.290Q-05
3	0.78693867696764847899565427329793Q+00	0.361Q-08	0.361Q-08
4	0.78693868057747578622098915156248Q+00	-0.274Q-11	-0.274Q-11
5	0.78693868057473174975633407525498Q+00	0.140Q-14	0.140Q-14
6	0.78693868057473315330773859861896Q+00	-0.515Q-18	-0.515Q-18
7	0.78693868057473315279225850921981Q+00	0.142Q-21	0.142Q-21
8	0.78693868057473315279240096069643Q+00	-0.307Q-25	-0.307Q-25
9	0.78693868057473315279240093001236Q+00	0.530Q-29	0.530Q-29
10	0.78693868057473315279240093001765Q+00	0.193Q-32	0.270Q-32
11	0.78693868057473315279240093001766Q+00	0.770Q-33	0.770Q-33
12	0.78693868057473315279240093001766Q+00	0.0	0.0

```
C     ..................................................................
C     .       THIS MAINLINE PROGRAM ILLUSTRATES THE SYNTAX OF THE SUB-   .
C     .  ROUTINE 'C1F1P' WHEN USED TO GENERATE COEFFICIENTS IN THE       .
C     .  POLYNOMIALS FOR THE PADE APPROXIMATION OF 1F1(1;CP;-Z).         .
C     ..................................................................
      IMPLICIT REAL*16(A-H,O-Z)
      DIMENSION CA(26),CB(26),NO(25)
 10   READ(5,2,END=999) CP
      WRITE(6,3) CP
      READ(5,1) M,(NO(J),J=1,M)
      DO 100 I=1,M
         N=NO(I)
C-----------------------------------------------------------------------
         CALL C1F1P(CP,CA,CB,N)
C-----------------------------------------------------------------------
C        IN THE ABOVE:
C
C  CP              IS THE DENOMINATOR PARAMETER OF THE 1F1 IN THE PADE
C                     APPROXIMATION
C  N               IS THE DEGREE OF THE POLYNOMIALS IN THE PADE
C                     APPROXIMATION
C  CA AND CB       WILL CONTAIN THE COEFFICIENTS IN THE NUMERATOR AND
C                     DENOMINATOR POLYNOMIALS , RESPECTIVELY
C
C        NOTE : THE COEFFICIENTS OF THE K-TH POWER OF Z WILL BE PLACED
C IN CA(K+1) OR CB(K+1) AS APPROPRIATE
C-----------------------------------------------------------------------
         N1=N+1
 100     WRITE(6,4) N,(CA(J),CB(J),J=1,N1)
      GOTO 10
 999  STOP
 1    FORMAT(26I2)
 2    FORMAT(Q39.32)
 3    FORMAT('1','COEFFICIENTS FOR THE PADE APPROXIMATION OF 1F1( 1 ; CP
     : ; -Z )'//' ',T20,'CP = ',Q39.32/)
 4    FORMAT(' ','N = ',I2,T18,'CA(I)',T58,'CB(I)'//
     :26(1X,Q39.32,2X,Q39.32/)/)
      END
```

```
      SUBROUTINE C1F1P(CP,A,B,N)
C     ************************************************************************
C     *       THIS SUBROUTINE RETURNS COEFFICIENTS A(I) AND B(I) ,           *
C     *  I = 1,2,...,N+1 , OF THE NUMERATOR AND DENOMINATOR POLYNOMIALS      *
C     *  RESPECTIVELY, IN THE PADE APPROXIMATION OF ORDER  N  FOR            *
C     *  1F1( 1 ; CP ; -Z ).                                                 *
C     *                                                                      *
C     *       NO OTHER SUBROUTINES ARE CALLED BY THIS ONE.                   *
C     ************************************************************************
      IMPLICIT REAL*16(A-H,O-Z)
      DIMENSION A(1),B(1)
      DATA ONE/1.Q0/,ZERO/0.Q0/
C
C  INITIALIZATION :
C
      XN=N
      XN1I=XN
      B(1)=ONE
      A(1)=ONE
      XI=ONE
      CP2NI=CP+XN+XN-ONE
      XIJ=ZERO
      DO 100 I=1,N
         I1=I+1
C-----------------------------------------------------------------------
C  FOR I = 1,2,...,N , B(I+1) IS COMPUTED AS FOLLOWS
C-----------------------------------------------------------------------
C
         B(I1)=XN1I/CP2NI*B(I)/XI
C
         A(I1)=ONE
         DO 50 J=1,I
C-----------------------------------------------------------------------
C  TO CALCULATE A(I+1), WE EMPLOY B(J), J = 1,2,...,I+1 AS FOLLOWS
C-----------------------------------------------------------------------
C
            A(I1)=B(J+1)-A(I1)/(CP+XIJ)
 50         XIJ=XIJ-ONE
         XIJ=XI
         XN1I=XN-XI
         CP2NI=CP2NI-ONE
 100     XI=XI+ONE
      RETURN
      END
```

COEFFICIENTS FOR THE PADE APPROXIMATION OF 1F1(1 ; CP ; -Z)

CP = 0.20000000000000000000000000000000Q+01

N = 3

CA(I)	CB(I)
0.10000000000000000000000000000000Q+01	0.10000000000000000000000000000000Q+01
-0.71428571428571428571428571428572Q-01	0.42857142857142857142857142857143Q+00
0.23809523809523809523809523809524Q-01	0.71428571428571428571428571428571Q-01
-0.11904761904761904761904761904761Q-02	0.47619047619047619047619047619047Q-02

N = 4

CA(I)	CB(I)
0.10000000000000000000000000000000Q+01	0.10000000000000000000000000000000Q+01
-0.55555555555555555555555555555556Q-01	0.44444444444444444444444444444444Q+00
0.27777777777777777777777777777778Q-01	0.83333333333333333333333333333333Q-01
-0.13227513227513227513227513227513Q-02	0.79365079365079365079365079365079Q-02
0.66137566137566137566137566137560Q-04	0.33068783068783068783068783068783Q-03

N = 5

CA(I)	CB(I)
0.10000000000000000000000000000000Q+01	0.10000000000000000000000000000000Q+01
-0.45454545454545454545454545454546Q-01	0.45454545454545454545454545454545Q+00
0.30303030303030303030303030303030Q-01	0.90909090909090909090909090909091Q-01
-0.12626262626262626262626262626263Q-02	0.10101010101010101010101010101010Q-01
0.12626262626262626262626262626264Q-03	0.63131313131313131313131313131313Q-03
-0.30062530062530062530062530062529Q-05	0.18037518037518037518037518037518Q-04

N = 6

CA(I)	CB(I)
0.10000000000000000000000000000000Q+01	0.10000000000000000000000000000000Q+01
-0.38461538461538461538461538461539Q-01	0.46153846153846153846153846153846Q+00
0.32051282051282051282051282051282Q-01	0.96153846153846153846153846153846Q-01
-0.11655011655011655011655011655011Q-02	0.11655011655011655011655011655012Q-01
0.17482517482517482517482517482517Q-03	0.87412587412587412587412587412587Q-03
-0.48562548562548562548562548562521Q-05	0.38850038850038850038850038850039Q-04
0.11562511562511562511562511562441Q-06	0.80937580937580937580937580937581Q-06